
Abstract

This paper presents a recent formalization of a Henkin-style completeness proof for the
propositional modal logic S5 using the Lean theorem prover [5]. The proof formalized is
close to that of Hughes and Cresswell [11], but the system, based on a different choice
of axioms, is better described as a Mendelson system augmented with axiom schemes
for K, T, S4, and B, and the necessitation rule as a rule of inference. The language
has the false and implication as the only primitive logical connectives and necessity as
the only primitive modal operator. The full source code is available online at https:
//github.com/bbentzen/mpl/ and has been typechecked with Lean 3.4.2.

1 Introduction
A proof of the completeness theorem for a given logic conforms to the Henkin style when it
applies nonconstructive methods to build models out of maximal consistent sets of formulas
(possibly after a Henkin language extension) using the deductive system itself. Henkin-style
completeness proofs for modal logics have been around for over five decades [14] but the formal
verification of completeness with respect to Kripke semantics is comparatively recent.

This paper presents a formalization of a Henkin-style completeness proof for the proposi-
tional modal logic S5 using the Lean theorem prover [5]. The proof is specific to S5, but, by
forgetting the appropriate extra accessibility conditions (as described in [11]), the technique we
use can be applied to weaker normal modal systems such as K, T, S4, and B. The formalization
covers the syntax and semantics of S5, syntactic and semantic deduction theorem, structural
rules (weakening, contraction, exchange), the recursive enumerability of the language, and
soundness and completeness results. In total, it has approximately 1,500 lines of code, but only
two thirds of it is required for the completeness proof. The full source code is available online
at https://github.com/bbentzen/mpl/ and has been typechecked with Lean 3.4.2. At the
time of writing, this is the latest stable release of Lean.

∗The author wishes to thank Jeremy Avigad, Mario Carneiro, Rajeev Goré, and Minchao Wu for helpful
suggestions and invaluable conversations on the topic covered herein. An early version of this development was
presented at the Lean Together 2019, Amsterdam, January 7–11, 2019.

https://github.com/bbentzen/mpl/
https://github.com/bbentzen/mpl/

1.1 Related work
The use of proof assistants in the mechanization of completeness proofs in the context of Kripke
semantics has been recently studied in the literature for a variety of formal systems. Coquand [3]
uses ALF to give a constructive formal proof of soundness and completeness with respect to
Kripke models for intuitionistic propositional logic with implication as the sole logical con-
stant. Building on Coquand’s work, a constructive completeness proof of Kripke semantics for
intuitionistic logic with implication and universal quantification has been verified with Coq by
Heberlin and Lee [10]. Also using Coq, Doczal and Smolka present a constructive formal proof
of completeness with respect to Kripke semantics and decidability of the forcing relation for an
extension of the basic modal logic K [6] and a variety of temporal logic [7]. In his formal verifi-
cation of cut elimination for coalgebraic logics, Tews [19] provides a formalization of soundness
and completeness proofs covering many different logics, including modal logic K.

However, to the best of our knowledge, the formalization of a Henkin-style completeness
proof for propositional modal logic S5 proposed in this paper is the first of its kind.1 Our proof
is close to that of Hughes and Cresswell [11], but given for a system based on a different choice
of axioms. In Hughes and Cresswell’s book, the basis of S5 is that of T plus an additional
axiom. Here S5 is built on axiom schemes for K, T, S4 and B. This has the advantage that we
can easily adapt the proof for different weaker systems. Another choice had to be made between
using a deep or a shallow embedding for the formalization. Because our aim is metatheoretical,
we use a deep embedding for the encoding of the syntax, as it allows us to prove metatheorems
by structural induction on formulas or derivations.

1.2 Lean
Lean is an interactive theorem prover developed principally by Leonardo de Moura and based
on a version of dependent type theory known as the Calculus of Inductive Constructions [16, 4].
Theorem proving in Lean can be done by constructing proof terms directly as in Agda [15], by
using tactics (imperative commands that describe how to construct a proof term) as in Coq [18],
or by mixing them together in the same proof environment. Like most proof assistants, Lean also
supports classical reasoning, which is employed in the formalization along with the declaration
of noncomputable constructions. Finally, the formalization also presupposes a few basic results
on data structures which are not in the standard library, so we make use of mathlib, the library
of formalized mathematics for Lean [2].

In the remainder of this paper, Lean code will be used to illustrate design choices and to give
a broad overview of the proof method, but not to discuss the proof itself. Interested readers
are encouraged to consult completeness.lean, the main file of the formalization where the
crucial steps of the proof are given in detail. We shall also give an informal proof sketch of the
completeness theorem using mathematical notation to convey the key ideas of the proof.

2 Modal logic
We start with the syntax and semantics of the modal logic we have formalized, namely, S5, the
strongest of the five systems of modal logic proposed by Lewis and Langford [12]. In the next
section, we give a proof sketch of completeness for the logic described here.

1In independent work done roughly at the same time the author first completed this formalization in 2018,
Wu and Goré [20] have described a formalization in Lean of modal tableaux for modal logics K, KT, and S4
with decision procedures with proofs of soundness and completeness. Also in 2018, but unknown to the author,
From [8] formalized a Henkin-style completeness proof for system K in Isabelle.

2.1 The language
For simplicity, we shall work with a language which has implication (⊃) and the false (⊥) as the
only primitive logical connectives, and necessity (2) as the only primitive modal operator. This
choice of primitive operators gives rise to modal logics with a complex lattice structure [13],
but it can also be used to do intuitionistic modal logic if one wishes to do so. This language
can be very conveniently defined using inductive types in Lean, in which case the parsing tree
of a formula is always made explicit—in fact, knowing that a certain formula is well-formed
amounts to knowing whether it is well-typed term of that type.
inductive form {σ :nat} : Type
| atom : fin σ → form
| bot : form
| impl : form → form → form
| box : form → form

Using one of the four constructors displayed above (atom, bot, impl, box) is the only way to
construct a term of type form. The elimination rule of this type is precisely the principle of
induction on the structure of a formula.

While newly-defined terms are always exhibited in Polish notation by default, Lean supports
unicode characters and has an extensible parser which allows the declaration of customized
prefix or infix notations for terms and types.
prefix `#` := form.atom
notation `⊥` := form.bot _
infix `⊃` := form.impl
notation `∼`:40 p := form.impl p (form.bot _)
prefix `2`:80 := form.box
prefix `3`:80 := λ p, ∼(2 (∼ p))

It is also possible to have notations for compound terms. Negation and possibility, for example,
are encoded as definable logical and modal constants. One of the benefits of having the false
as primitive rather than negation is that the statement that a certain set of premises (context)
is consistent need not be dependent on particular choices of variables. In other words, we can
write Γ ⊬S5 ⊥ instead of Γ ⊬S5 p & ∼p to express that Γ is consistent.

2.2 Contexts
Contexts are just sets of formulas, so, since we already have a type of formulas, implementing
contexts does not require extra work. However, as there is a variety of set-like objects in Lean
(lists, multisets, finsets, sets etc.), there are many possible design choices.

In an earlier version of the formalization, contexts were implemented using lists. A strong
point of this approach is that one can reason by induction on contexts (lists are inductive
types), allowing, among other things, a reduction of strong completeness to weak completeness
via the deduction theorem [1]. Unfortunately, since lists must always be finite in size, there is
no consistent way of reproducing maximal consistent extensions of contexts this way because
every maximal consistent set is necessarily infinite.

Contexts are now defined using sets, which are functions of type A → Prop. Simply put,
a set is just a predicate, a non-ordered collection that cannot contain duplicates (as a result,
exchange and contraction are trivial structural rules of the proof system). For the purpose of
stating more readable functions which look like logic textbook theorems, we have:
def ctx : Type := set (form σ)

notation `·` := {}

notation Γ `̀ ` p := set.insert p Γ

2.3 The proof system
We define a Hilbert-style axiomatic system, which is presented as a Mendelson system aug-
mented with axiom schemes for K, T, S4, and B, and the necessitation rule as rule of inference.
Such a proof system can be succinctly defined as follows in Lean:
inductive prf : ctx σ → form σ → Prop
| ax {Γ} {p} (h :p∈ Γ): prf Γ p
| pl1 {Γ} {p q} : prf Γ (p ⊃ (q ⊃ p))
| pl2 {Γ} {p q r} : prf Γ ((p ⊃ (q ⊃ r)) ⊃ ((p ⊃ q) ⊃ (p ⊃ r)))
| pl3 {Γ} {p q} : prf Γ (((∼p) ⊃ ∼q) ⊃ (((∼p) ⊃ q) ⊃ p)
| k {Γ} {p q} : prf Γ (2(p ⊃ q) ⊃ (2p ⊃ 2q))
| t {Γ} {p} : prf Γ (2p ⊃ p)
| s4 {Γ} {p} : prf Γ (2p ⊃ 22p)
| b {Γ} {p} : prf Γ (p ⊃ 23p)
| mp {Γ} {p q} (hpq: prf Γ (p ⊃ q)) (hp :prf Γ p) :prf Γ q
| nec {Γ} {p} (h :prf · p) : prf Γ (2p)

notation Γ ` ⊢s5 ` p := prf Γ p
notation Γ ` ⊬s5 ` p := prf Γ p → false

Note that the use of the empty context · in the definition of nec (the necessitation rule) is
essential to the completeness proof described in this paper. We shall return to it soon when
discussing the deduction theorem and semantic consequence relation.

But before addressing those, it is worth pausing for a moment to give an illustration of how
terms of the type of proofs defined above can be used to represent actual derivations in the
modal logic S5. In particular, notice that the following term
lemma box_dne {p : form σ} :
· ⊢S5 (2∼∼p) ⊃ (2p) :=
prf.mp (prf.k) (prf.nec (prf.dne))

translates as a proof tree of the following theorem

2p ⊃ 2∼∼p

(mp)2(p ⊃ ∼∼p) ⊃ (2p ⊃ 2∼∼p)

(k)

2(p ⊃ ∼∼p)

(nec)p ⊃ ∼∼p

(dne)

.

In some deductive systems for modal logic, the deduction theorem, which states that if a
formula q is deducible from a set of assumptions Γ∪{p} then the implication p ⊃ q is deducible
from Γ, does not hold. In this formulation, it can be proven by a straightforward induction on
the structure of the proof tree or, what amounts to the same thing, by an application of the
elimination rule of the type of proofs defined above.
theorem deduction {Γ :ctx σ} {p q :form σ} :
(Γ ` p ⊢s5 q) → (Γ ⊢s5 p ⊃ q) :=
begin
generalize eq : (Γ ` p) =Γ',
intro h,
induction h; subst eq,
{ repeat {cases h_h},

exact id,
{ exact mp pl1 (ax h_h) } },
{ exact mp pl1 pl1 },
{ exact mp pl1 pl2 },
{ exact mp pl1 pl3 },
{ apply mp,
{ exact (mp pl2 (h_ih_hpq rfl)) },
{ exact h_ih_hp rfl } },
{ exact mp pl1 k },
{ exact mp pl1 t },
{ exact mp pl1 s4 },
{ exact mp pl1 b },
{ exact mp pl1 (nec h_h) }
end

The proof of all cases follows from a simple application of modus ponens to axiom scheme
(1), except for the modus ponens case that requires axiom scheme (2). The presence of the
necessitation rule in the form of nec does not impose any difficulty in the proof because the
context restriction in the antecedent gives us a stronger inductive hypothesis.

As noted in [9], whether the deduction theorem fails or not in modal logics which have the
necessitation rule is a consequence of design choice. The necessitation rule, which informally
says that if p is a theorem of the system then so is 2p, can be implemented in many ways. As
can be seen above, we restrict its applicability to theorems by only allowing formulas that are
derivable under no assumptions in the antecedent of nec. Another possibility is having a more
general (i.e. unrestricted) formulation of the necessitation rule, which extends the range of
application of nec to formulas that are derivable under assumptions. That would mean having
a constructor such as nec' instead of nec in the formalization:
| nec' {Γ} {p} (h :prf Γ p) :prf Γ (2p)

while this rule can be useful and, indeed, the resulting system can be shown to be complete
with respect to a global version of semantic consequence [17] (see section 2.4.2), the deduction
theorem does not hold in the presence of it.

2.4 Semantics
2.4.1 Kripke models

The semantics for S5 are given by Kripke semantics. A model M is a triple ⟨W,R, v⟩ where

• W is a non-empty set of objects called possible worlds;

• R is a binary, equivalence relation on possible worlds;

• v specifies the truth value of a formula at a world.

Typically, there are no constraints on what kind of objects the members of W (possible worlds)
should be. For our purposes, however, it is useful to let them be sets of formulas instead of
arbitrary objects (there is no loss of generality in the resulting semantics). We thus have:
def wrld (σ :nat) := set (form σ)

Kripke models can be implemented as structures (inductive types with only one constructor).
This can be done using the structure command in Lean. In the following we define a 6-tuple
composed of a domain, an accessibility relation, a valuation function, and reflexivity, symmetry
and transitivity proofs for the given relation.

structure model :=
(wrlds : set (wrld σ))
(access : wrld σ → wrld σ → bool)
(val : fin σ → wrld σ → bool)
(refl : ∀ w ∈ wrlds, access w w = tt)
(symm : ∀ w ∈ wrlds, ∀ v ∈ wrlds, access w v = tt → access v w = tt)
(trans : ∀ w ∈ wrlds, ∀ v ∈ wrlds, ∀ u ∈ wrlds,
access w v = tt → access v u = tt → access w u = tt)

The Boolean type bool is used in the formalization of truth values (i.e. either tt or ff).

2.4.2 Semantic consequence

We have a recursively defined forcing function which takes a model, a formula, and a world
as inputs and returns a boolean value. It can be defined by induction on the structure of the
formula involved as follows:
def forces_form (M : model) : form σ → wrld σ → bool
| (#p) := λ w, M.val p w
| (bot σ) := λ w, ff
| (p ⊃ q) := λ w, (bnot (forces_form p w)) | | (forces_form q w)
| (2p) := λ w, if (∀ v ∈ M.wrlds, w ∈ M.wrlds →
M.access w v = tt → forces_form p v = tt) then tt else ff

The definition is a routine adaptation of that found in traditional modal logic textbooks [11].
Non-modal connectives are given truth-functionally and the necessity operator is defined by
stating that a formula 2p is true at a world w iff if R(w, v) then p is true at v, for all v ∈ W.

This function can be extended to contexts in the obvious way: we say that a context is true
at a world in a model if each formula of that context is true at that world and in that model.
def forces_ctx (M : model) (Γ :ctx σ) :wrld σ → bool :=
λ w, if (∀ p, p ∈ Γ → forces_form M p w = tt) then tt else ff

A formula p is a local semantic consequence of a context Γ iff, for all models M and for all
worlds w ∈ W, the fact that Γ is true at w in M implies that p is true at w in M.
inductive sem_csq (Γ :ctx σ) (p :form σ) :Prop
| is_true (m : ∀ (M :model) (w : wrld σ),
forces_ctx M Γ w = tt → forces_form M p w = tt) : sem_csq

notation Γ `⊨s5` p := sem_csq Γ p

Say that a formula (context) is true in a model M if that formula (context) is true at all worlds
in that model. A formula p is a global semantic consequence of a context Γ iff, for any model M,
if Γ is true in M then p is true in M. It is easy to see that local semantic consequence implies
global semantic consequence, so the global version of completeness implies the local version.
The converse does not hold. In particular, note that 2p is a global—but not a local—semantic
consequence of {p}. The formalization includes local but not global semantic consequence. It
is not hard to encode, but the global version of completeness is actually false for our proof
system.2

3 The completeness theorem
In this section we formalize a proof of completeness with respect to the proof system and
semantics developed in the previous sections. First, we state the completeness theorem. Second,

2Unless nec is replaced with nec'. See [17] for an informal proof.

a general outline of the proof is presented. Next, we give an informal explanation of each
individual proof step followed by a description of its respective implementation.

Theorem 3.0.1 (Completeness). For every context Γ, any formula p that follows semantically
from Γ is also derivable from Γ in the modal logic S5. In symbols:

Γ ⊨S5 p =⇒ Γ ⊢S5 p

That is, every semantic consequence is also a syntactic consequence in S5.

The proof here requires full contraposition and it is thus non-constructive. More specifically,
the idea is to assume that both Γ ⊨S5 p and Γ ⊬S5 p hold, and then derive a contradiction using
the syntax to build a model M = ⟨W,R, v⟩ (the canonical model) where Γ is true but p is
false at a specific world in the domain. To complete the proof we shall need some additional
lemmas, many of which can be easily proven by induction.

We shall focus on sketching the formal argument for the following facts:

1. Γ ∪ {∼p} has a maximal consistent extension ∆ defined as follows:

∆0 :=Γ ∪ {∼p}

∆n+1 :=

{
∆n ∪ {φn+1} if ∆n ∪ {φn+1} is consistent
∆n ∪ {∼φn+1} otherwise

∆ :=
∪
n∈N

∆n

(i.e. ∆ is consistent, maximal and Γ ∪ {∼p} ⊆ ∆)

2. There exists a canonical model where p is true at w iff p ∈ w;

3.0.1 Maximal consistent sets

We say that a context is maximal consistent if it is consistent and, moreover, for every formula
expressible in the language, either it or its negation is contained in that context.
def is_max (Γ :ctx σ) := is_consist Γ ∧ (∀ p, p ∈ Γ ∨ (∼p) ∈ Γ)

As our language is countable, it is possible to construct each ∆n+1 using natural numbers to run
through the set of all formulas, deciding whether or not a number’s corresponding formula (when
it exists) is consistent with ∆n or not (another alternative is to generalize this construction to
languages of arbitrary cardinalities using Zorn’s lemma instead).

The enumerability of the language is expressed using encodable types, which are construc-
tively countable types. Essentially, a type α is encodable when it has an injection encode :α →
nat and a (partial) inverse decode :nat → option α.

def insert_form (Γ :ctx σ) (p :form σ) :ctx σ :=
if is_consist (Γ ` p) then Γ ` p else Γ ` ∼p

def insert_code (Γ :ctx σ) (n :nat) : ctx σ :=
match encodable.decode (form σ) n with
| none := Γ
| some p := insert_form Γ p
end

def maxn (Γ :ctx σ) :nat → ctx σ

| 0 := Γ
| (n+1) := insert_code (maxn n) n

def max (Γ :ctx σ) :ctx σ :=∪
n, maxn Γ n

Before proceeding any further, we must show that max Γ is the maximal consistent extension
of Γ, that is, that Γ in contained in max Γ and that max Γ is maximal and consistent.

First, we note that, for each maxn Γ n of the family of sets, we have Γ ⊆ maxn Γ n. So Γ
must also be contained in their union, max Γ. This proof argument produces a term of type:
lemma subset_max_self {Γ :ctx σ} :
Γ ⊆ max Γ

Second, we observe that every formula must be in the enumeration somewhere, so suppose that
the formula p has index i. By the definition of maxn Γ i, either p or ∼p is a member of maxn
Γ i, so one of them is a member of max Γ. Thus, we have a term
theorem mem_or_mem_max {Γ :ctx σ} (p :form σ) :
p ∈ max Γ ∨ (∼p) ∈ max Γ

Third, assume for the sake of contradiction that Γ is consistent but max Γ is not. By structural
induction on the proof tree, we prove that there exists an i such that maxn Γ i is inconsistent.
However, each maxn Γ i preserves consistency. This gives a function
lemma is_consist_max {Γ :ctx σ} :
is_consist Γ → is_consist (max Γ)

The above proof sketches are implemented purely by unfolding definitions and inductive
reasoning. They consist of approximately 150 lines of code in completeness.lean. There
is even a one-line short case-reasoning proof that maximal consistent sets are closed under
derivability:
lemma mem_max_of_prf {Γ :ctx σ} {p :form σ} (h1 :is_max Γ)
(h2 :Γ ⊢S5 p) :p ∈ Γ :=
(h1.2 p).resolve_right (λ hn, h1.1 (prf.mp (prf.ax hn) h2))

3.0.2 The canonical model construction

In this section we build the canonical model for S5 from the deductive system itself. For that
we specify a domain, an accessibility relation, a valuation function, and proofs of reflexivity,
symmetry and transitivity for the accessibility relation. We build the model as follows:

• W is the set of all maximal consistent sets of formulas;

• R(w, v) iff 2p ∈ w implies p ∈ v;

• v(w, p) = 1 if w ∈ W and p ∈ w, for a propositional letter p.

For that to be a well-defined model, we must to show that R is an equivalence relation.
Reflexivity translates as follows: 2p ∈ w implies p ∈ w for a given world w ∈ W. But this

is easy because w is closed under derivability (it is a maximal consistent set of formulas) and
our proof system has modus ponens and axiom schema (t).

Proving symmetry requires more work. Given any worlds w, v ∈ W, suppose first that
2φ ∈ w implies φ ∈ v for all formulas φ, and suppose that 2p ∈ v. We want to show that
p ∈ w. Since 32p ⊃ p is a theorem of S5 (see syntax/lemmas.lean) we just have to prove

that 32p ∈ w, or, equivalently, that 2∼2p /∈ w. By contraposition on our initial hypothesis,
it suffices to show that ∼2p /∈ v. But 2p ∈ v and v is consistent.

For transitivity, we must show that p ∈ u, on the assumptions that 2p ∈ w, that 2φ ∈ w
implies φ ∈ v, and that 2φ ∈ v implies φ ∈ u, for any formula φ. In other words, we want to
show that 22p ∈ w. But this follows from modus ponens and axiom scheme (s4).

This model construction is represented by the Lean code
def domain (σ :nat) : set (wrld σ) := {w | ctx.is_max w}

def unbox (w : wrld σ) :wrld σ := {p | (2p) ∈ w}

def access : wrld σ → wrld σ → bool :=
λ w v, if (unbox w ⊆ v) then tt else ff

def val : fin σ → wrld σ → bool :=
λ p w, if w ∈ domain σ ∧ (#p) ∈ w then tt else ff

lemma mem_unbox_iff_mem_box {p : form σ} {w :wrld σ} :
p ∈ unbox w ↔ (2p) ∈ w :=
⟨ id, id ⟩

What is here called unbox is a set operation which takes a set of formulas w as an input and
returns the set of formulas p such that 2p is a member of w.

A rather trivial but still quite useful lemma about this operation is that if p is deducible
from unbox w then actually 2p ∈ w. It can be proved by structural induction on the derivation
using the necessitation rule. The proof argument is straightforward but, because we shall need
this fact in the next section, it is convenient to have the formal proof presented here:
lemma mem_box_of_unbox_prf {p : form σ} {w :wrld σ}
(H : w ∈ domain σ) :(unbox w ⊢s5 p) → (2p) ∈ w :=
begin
generalize eq : unbox w = Γ',
intro h, induction h; subst eq,
{ assumption },
repeat { apply ctx.mem_max_of_prf H,
apply prf.nec,
apply prf.pl1 < | > apply prf.pl2 < | > apply prf.pl3 },
{ apply ctx.mem_max_of_prf H,
refine prf.mp (prf.ax _) (prf.ax (h_ih_hp rfl)),
exact (ctx.mem_max_of_prf H)
(prf.mp prf.k (prf.ax (h_ih_hpq rfl))) },
{ apply ctx.mem_max_of_prf H,
exact prf.nec prf.k },
{ apply ctx.mem_max_of_prf H,
exact prf.nec prf.t },
{ apply ctx.mem_max_of_prf H,
exact prf.nec prf.s4 },
{ apply ctx.mem_max_of_prf H,
exact prf.nec prf.b },
{ apply ctx.mem_max_of_prf H,
apply prf.nec (prf.nec h_h) }
end

3.0.3 Truth and membership

In order to prove completeness, we first show that truth is membership in the canonical model,
that is, that a formula is true at a world in the canonical model iff it is a member of that world:

lemma form_tt_iff_mem_wrld {p : form σ} :
∀ (w ∈ domain σ), (forces_form model w p) = tt ↔ p ∈ w

where model is the canonical model defined in the previous section. To prove this, we use
induction on the structure of the formula p. So, we will show that the result holds when p is a
propositional letter #p, the false ⊥, an implication p ⊃ q or a necessary proposition 2p.

In the proof mechanization, we use the induction tactic in the tactic mode. This tactic
produces four goals (as displayed below).
case form.atom
σ :N,
p : fin σ
⊢ ∀ (w :wrld σ),
w ∈ domain σ → forces_form model w #p = tt ↔ #p ∈ w)

The first goal is a base case. If p is a propositional letter the biconditional statement is clearly
true, because, by definition of the valuation function, this is what it means for a propositional
letter to be true at a world in the canonical model.
case form.bot
σ :N,
⊢ ∀ (w :wrld σ),
w ∈ domain σ → forces_form model w ⊥ = tt ↔ ⊥ ∈ w)

Proving the second goal is easy. Both conditional statements are vacuously true in both di-
rections for the false, because w is consistent and, by definition, the false cannot be true at a
world in a model.
case form.impl
σ :N,
p q : form σ,
hp :
∀ (w :wrld σ),
w ∈ domain σ → forces_form model w p = tt ↔ p ∈ w),
hq :
∀ (w :wrld σ),
w ∈ domain σ → forces_form model w q = tt ↔ q ∈ w)
⊢ ∀ (w :wrld σ),
w ∈ domain σ → forces_form model w (p ⊃ q) = tt ↔ (p ⊃ q) ∈ w)

For the third goal, we assume the inductive hypothesis that the result holds for p and q, and
then show that either p is false or q is true at w in the canonical model iff p ⊃ q ∈ w. The proof
follows from case reasoning and the closure under derivability of possible worlds.
case form.box
σ :N,
p : form σ,
hp :
∀ (w :wrld σ),
w ∈ domain σ → forces_form model w p = tt ↔ p ∈ w)
⊢ ∀ (w :wrld σ),
w ∈ domain σ → forces_form model w (2p) = tt ↔ (2p) ∈ w)

To prove the fourth goal, we begin by assuming the inductive hypothesis for p. If w is a world,
and, if it is a maximal consistent set of formulas, then, by unfolding the definition of truth of
a formula at a world in a model, the biconditional statement becomes
⊢ (∀ (v :wrld σ),
v ∈ model.wrlds → w ∈ model.wrlds → model.access w v = tt → forces_form model w p =

tt) ↔ (2p) ∈ w)

10

In the forwards direction, we assume that 2p is true at w in the canonical model and
that ∼2p ∈ w. But then, by lemma mem_box_of_unbox_prf, the context unbox w ∪ {∼p} is
consistent and can be extended to a maximal consistent set (i.e. a world in the domain). It is
accessible to w because unbox w ⊆ max (unbox w ∪ {∼ p}), so p should be true at w. But p /∈
max (unbox w ∪ {∼ p}) because it is consistent.

For the backwards direction, assume that 2p ∈ w. Given a maximal consistent set of
formulas v and assuming that 2φ ∈ w then φ ∈ v for all φ, we have to show that p is true at
v in the model. By the inductive hypothesis, however, it suffices to show that p ∈ v, but this
follows from 2p ∈ w.

3.0.4 The completeness proof

We now complete our proof by putting together all the above pieces into 24 lines of code. Since
we know by hypothesis that Γ ⊬S5 p, it follows that Γ ∪ {∼p} is consistent—otherwise, if the
false were deducible from it, we would have a contradiction by double negation elimination.
lemma consist_not_of_not_prf {Γ :form σ} {p :form σ} :
(Γ ⊬s5 p) → is_consist (Γ ` ∼p) :=
λ hnp hc, hnp (mp dne (deduction hc))

Now assuming that Γ ⊨S5 p, the basic idea for deriving the contradiction is that, as max Γ∪{∼p}
is a world in the canonical model, and each formula φ ∈ Γ is true at that world, Γ is true as
well. Clearly, p is not consistent with Γ ∪ {∼p}, so p /∈ max Γ ∪ {∼p}, meaning that p must be
false at that world.

This allows us to prove the following theorem
theorem completenss {Γ :form σ} {p :form σ} :
(Γ ⊨s5 p) → Γ ⊢s5

4 Conclusion
We presented a computer formalization of the completeness of the modal logic S5 using a
Henkin-style argument, as given in traditional textbooks such as [11]. An interesting direction
for further work would be to formally verify the global version of completeness for the alternative
presentation of S5 obtained by replacing nec with nec'. Another natural step for continuing
the work presented here is to present completeness proofs for different modal logics.

